式の変形(I, simplify)

数式の変形は，手で直すほうが圧倒的に早くきれいになる場合が多い．しかし，テイラー展開や，複雑な積分公式，三角関数とexp関数の変換などの手間がかかるところを，Mapleは間違いなく変形してくれる．ここで示すコマンドを全て覚える必要は全くない．というか忘れるもの．ここでは，できるだけコンパクトにまとめて，悩んだときに参照できるようにする．初めての人は，ざっと眺めた後，鉄則からじっくりフォローせよ．

解説

コマンドの分類

まず数式処理でよく使うコマンドをいくつかの範疇に分類してまとめておく．このほかにも前に示した，solve(解), diff(微分), int(積分)，series(級数展開)等は頻繁に数式の導出・変形に登場する．

式の変形	式の分割抽出	代入，置換，仮定	省略操作，その他
simplify:簡単化 expand:展開 factor:因数分解 normal:約分・通分 combine:公式でまとめる collect:次数でまとめる sort:昇べき，降べき convert:形式の変換	lhs, rhs:左辺，右辺 numer, denom:分子，分母 coeff:係数 nops, op	subs:一時的代入 asume:仮定 assuming:一時的仮定 assign:値の確定 about:仮定の中身 anames('user'):使用変数名 restart,a:='a':初期化	||:連結作用素 seq:for-loopの簡易表記 map:関数の要素への適用 add,mul:単純な和，積 sum,product:数式に対応した和，積 limit:極限

式の変形に関連したコマンド

simplify(exp1):簡単化> simplify(exp1,副関係式):

```
> simplify(3*x+4*x+2*y);
```

$$7x+2y \quad (1.1.2.1.1)$$

```
> exp1:=3*sin(x)^3-sin(x)*cos(x)^2;
  simplify(exp1);
```

$$exp1 := 3\sin(x)^3 - \sin(x)\cos(x)^2$$
$$-\left(4\cos(x)^2-3\right)\sin(x) \quad (1.1.2.1.2)$$

```
> simplify(exp1,{cos(x)^2=1-sin(x)^2});
```

$$4\sin(x)^3-\sin(x) \quad (1.1.2.1.3)$$

オプションとしてsizeを指定するとより簡単になる場合がある．

```
> simplify(exp1,size):
```

expand(exp1):展開

```
> expand((x+y)^2);
```

$$x^2+2xy+y^2 \quad (1.1.2.2.1)$$

factor(exp1):因数分解

```
> factor(4*x^2-6*x*y+2*y^2);
```

$$2(2x-y)(x-y) \quad (1.1.2.3.1)$$

normal(exp1):約分・通分

```
> normal((x+y)/(x^2-3*x*y-4*y^2));
```

$$\frac{1}{x-4y} \quad (1.1.2.4.1)$$

```
> normal(1/x+1/y);
```

$$\frac{y+x}{xy} \quad (1.1.2.4.2)$$

collect(exp1,exp2):項を変数でまとめる

```
> collect(4*a*x^2-3*y^2/x+6*b*x*y+3*c*y+2*y^2,y);
```

$$\left(-\frac{3}{x}+2\right)y^2+(6bx+3c)y+4ax^2 \quad (1.1.2.5.1)$$

combine(exp1):項を公式でまとめる

```
> combine(sin(x)^2+3*cos(x)^2);
```

$$2+\cos(2x) \quad (1.1.2.6.1)$$

sort(exp1):昇べき，降べき

> sort(exp1,[x,y]);
> sort(exp1, [x], opts);opts=tdeg(総次数順),plex(辞書式順),ascending(昇順), descending(降順)

```
> exp1:=x^3+4*x-3*x^2+1:
  sort(exp1);
```

$$x^3-3x^2+4x+1 \quad (1.1.2.7.1)$$

```
> sort(exp1,[x],ascending);
```

$$1+4x-3x^2+x^3 \quad (1.1.2.7.2)$$

```
> exp2:=x^3-a*x*y+4*x^2+y^2:
  sort(exp2);
```

$$-axy+x^3+4x^2+y^2 \quad (1.1.2.7.3)$$

```
> sort(exp2,[x]);
```

$$x^3+4x^2-ayx+y^2 \quad (1.1.2.7.4)$$

```
> sort(exp2,[a,y]);
  sort(exp2,[a],plex);
```

$$-xay+y^2+x^3+4x^2$$
$$-xya+y^2+x^3+4x^2 \quad (1.1.2.7.5)$$

演習

1. 以下の式を簡単化せよ．
i) x^100-1 ii) x^2-y^2+2*x+1 iii) (a+b+c)^3-(a^3+b^3+c^3)

式の変形(II, convert, 分割抽出)

解説

式の変換

convert(exp1,opt):形式の変換

opt	意味
polynom	級数を多項式(polynomial)に
trig	三角関数(trigonal)に
sincos	tanを含まない、sin,cosに
exp	指数関数形式に
parfrac	部分分数(partial fraction)に
rational	浮動小数点数を有理数形式に

```
> s1:=series(sin(x),x,4);
  convert(s1,polynom);
```

$$s1 := x - \frac{1}{6}x^3 + O(x^4)$$

$$x - \frac{1}{6}x^3 \qquad (2.1.1.1.1)$$

```
> convert(sin(x),exp);
```

$$-\frac{1}{2}I\left(e^{Ix} - e^{-Ix}\right) \qquad (2.1.1.1.2)$$

```
> convert(sinh(x),exp);
```

$$\frac{1}{2}e^{x} - \frac{1}{2}e^{-x} \qquad (2.1.1.1.3)$$

```
> convert(tan(x),sincos);
```

$$\frac{\sin(x)}{\cos(x)} \qquad (2.1.1.1.4)$$

```
> convert(exp(I*x),trig);
```

$$\cos(x) + I\sin(x) \qquad (2.1.1.1.5)$$

```
> convert(1/(x-1)/(x+3),parfrac);
```

$$-\frac{1}{4(x+3)} + \frac{1}{4(x-1)} \qquad (2.1.1.1.6)$$

```
> convert(3.14,rational);
```

$$\frac{157}{50} \qquad (2.1.1.1.7)$$

式の分割・抽出に関連したコマンド

lhs(exp1), rhs:左辺，右辺

```
> lhs(sin(x)^2=1-1/x);
  rhs(sin(x)^2=1-1/x);
```

$$\sin(x)^2$$

$$1 - \frac{1}{x} \qquad (2.1.2.1.1)$$

numer(exp1),denom:分子，分母

```
> numer(a*x/(x+y)^3);
  denom(a*x/(x+y)^3);
```

$$a\,x$$

$$(x+y)^3 \qquad (2.1.2.2.1)$$

coeff(exp1,x^2):係数

```
> coeff(4*a*x^2-3*y^2/x+6*b*x*y+3*c*y+2*y^2,y^2);
```

$$-\frac{3}{x} + 2 \qquad (2.1.2.3.1)$$

op(exp1), nops(exp1):要素の取りだし，要素数

op, nopsはlist配列から要素や要素数を取り出すのに便利だが，より一般的な構造に対しても作用させることができる．

```
> op(4*a*x^2-3*y^2/x+6*b*x*y+3*c*y+2*y^2);
```

$$4\,a\,x^2,\ -\frac{3\,y^2}{x},\ 6\,b\,x\,y,\ 3\,c\,y,\ 2\,y^2 \qquad (2.1.2.4.1)$$

```
> nops(4*a*x^2-3*y^2/x+6*b*x*y+3*c*y+2*y^2);
```

$$5 \qquad (2.1.2.4.2)$$

演習

1. 以下の関数をx0まわりで３次までテイラー展開し，得られた関数ともとの関数をプロットせよ．さらに高次まで展開した場合はどう変化するか．
i) y=cos(x), x0=0　ii) y=ln(x), x0=1　iii) y=exp(-x), x0=0

2. $\frac{x+1}{(x-1)(x^2+1)^2}$ を部分分数に展開せよ．

3. $\frac{1}{1-x^4} = \frac{a}{(x^2+1)} + \frac{b}{(x+1)} + \frac{c}{(x-1)}$ が常に成立するa, b, cを定めよ．

4. $\frac{8}{3-\sqrt{5}} - \frac{2}{2+\sqrt{5}}$ を簡単化せよ．

5. $x^2 + 2kx + (5-k)=0$が重根をもつようにkを定めよ．

式の変形(III, assume, subs)

解説

代入，置換，仮定に関連したコマンド

subs(関係式, exp1):一時的代入

```
> exp1:=x^2-4*x*y+4;
  subs(x=2,exp1);
```

$$exp1 := x^2 - 4\,x\,y + 4$$

$$8 - 8\,y \tag{3.1.1.1.1}$$

```
> subs({x=a+2,y=sin(x)},exp1);
```

$$(a+2)^2 - 4\,(a+2)\,\sin(x) + 4 \tag{3.1.1.1.2}$$

assume(関係式):仮定

```
> sqrt(b^2);
  assume(a>0);
  sqrt(a^2);
```

$$\sqrt{b^2}$$

$$a\sim \tag{3.1.1.2.1}$$

exp1 assuming 関係式:一時的仮定

```
> exp1:=x^2-4*x+4;
  sqrt(exp1);
```

$$exp1 := x^2 - 4\,x + 4$$

$$\sqrt{(-2+x)^2} \tag{3.1.1.3.1}$$

```
> sqrt(exp1) assuming x>2;
```

$$-2 + x \tag{3.1.1.3.2}$$

additionally:assumeに加えての仮定

assign(exp1):solveで求めた値の確定

```
> x:='x':y:='y':
  s1:=solve({x-y+1=0,x+y-2=0},{x,y});
  assign(s1);
```

$$s1 := \left\{y = \frac{3}{2},\, x = \frac{1}{2}\right\} \tag{3.1.1.5.1}$$

```
> x,y;
```

$$\frac{1}{2},\,\frac{3}{2} \tag{3.1.1.5.2}$$

about:assumeで仮定した内容の確認

```
> about(a);
Originally a, renamed a~:
  is assumed to be: RealRange(Open(0),infinity)
```

anames('user'):ユーザが定義した変数の確認

```
> anames('user');
```

$$s1,\, y,\, x,\, a \tag{3.1.1.7.1}$$

restart,a:='a': 値の初期化

省略操作，その他のコマンド

||:連結作用素，前後の変数をくっつけて新たな変数とする．

```
> a||1;
  a||b;
```

$$a1$$

$$ab \tag{3.1.2.1.1}$$

```
> for i from 1 to 3 do
    a||i:=i^2;
  end do:
```

seq(exp1,i=0..3):for-loopの単純表記

```
> seq(k,k=4..7);
```

$$4, 5, 6, 7 \tag{3.1.2.2.1}$$

map(exp1,i=0..3):リスト要素への関数の一括適用

```
> f:=x->exp(-x);
  map(f,[0,1,2,3]);
```

$$f := x \to e^{-x}$$

$$[1, e^{-1}, e^{-2}, e^{-3}] \tag{3.1.2.3.1}$$

上記の３つを組み合わせると，効率的に式を扱うことができる．

```
> map(sin,[seq(theta||i,i=0..3)]);
```

$$[\sin(\theta 0), \sin(\theta 1), \sin(\theta 2), \sin(\theta 3)] \tag{3.1.2.3.2}$$

add,mul:単純な和，積

```
> add(x^i,i=0..3);
```

$$1 + x + x^2 + x^3 \tag{3.1.2.4.1}$$

```
> mul(x^i,i=0..3);
```

$$x^6 \tag{3.1.2.4.2}$$

sum,product:数式に対応した和，積

```
> add(x^i,i=0..n);
Error, unable to execute add
> sum(x^i,i=0..n);
```

$$\frac{x^{n+1}}{x-1} - \frac{1}{x-1} \tag{3.1.2.5.1}$$

```
> product(x^i,i=0..n);
```

$$x^{\frac{1}{2}(n+1)^2 - \frac{1}{2}n - \frac{1}{2}} \tag{3.1.2.5.2}$$

limit:極限

```
> limit(exp(-x),x=infinity);
```

$$0 \tag{3.1.2.6.1}$$

```
> limit(tan(x),x=Pi/2);
```

$$undefined \tag{3.1.2.6.2}$$

```
> limit(tan(x),x=Pi/2,left);
  limit(tan(x),x=Pi/2,complex);
```

$$\infty$$

$$-\infty + \infty\,I \tag{3.1.2.6.3}$$

式の変形(IV, 基本と奥の手)

解説

どうしても解かなければならない課題を前にコマンドリファレンスのあちこちを参照しながら解いていくのが数式処理を修得する最速法である．とびかかる前にちょっとした共通のコツがある．それをここでは示す．数式処理ソフトでの数式処理とは，数式処理ソフトが『自動的にやって』くれるのではなく，実際に紙と鉛筆で解いていく手順を数式処理ソフトに『やらせる』ことであることを肝に銘じよ．

鉄則

Mapleをはじめとする数式処理ソフトの習得にあたって初心者がつまづく共通の過ちを回避する鉄則がある．

鉄則０：restart をかける

続けて入力すると前の入力が生きている．違う問題へ移るときや，もう一度入力をし直すときには，restart;を入力して初期状態からはじめる．入力した順番が狂っている場合もある．頭から順にreturnをやり直す．

鉄則１：出力してみる

多くのテキストではページ数の関係で出力を抑止しているが，初心者が問題を解いていく段階ではデータやグラフをできるだけ多く出力する．最後のコロンをセミコロンに変える，あるいは途中にprint文を入れる．

鉄則２：関数に値を代入してみる

数値が返ってくるべき時に変数があればどこかで入力をミスっている．plotで以下のようなエラーが出た場合にチェック．

```
> plot(f(x),x);
Warning, unable to evaluate the function to numeric
values in the region; see the plotting command's help
page to ensure the calling sequence is correct
```

鉄則３：内側から順に入力する

長い入力やfor-loopを頭から打ち込んではいけない！！　内側から順に何をしているか解読・確認しながら打ち込む．括弧が合わなかったり，読み飛ばしていたりというエラーが回避できる．

具体例：無限積分

以下に示す積分を実行せよ．

$$\int_{-\infty}^{\infty} x e^{-\beta c x^2}\left(1+\beta g x^3\right) \mathrm{d}x$$

最新版のMapleでは改良が施されていて，このような複雑な積分も一発で求まる．

```
> f1:=unapply(x*exp(-beta*c*x^2)*(1+beta*g*x^3),x);
```

$$f1 := x \rightarrow x\,\mathrm{e}^{-\beta c x^2}\left(1+\beta g x^3\right) \tag{4.1.2.1}$$

```
> int(f1(x),x=-infinity..infinity);
```

$$\begin{cases} \frac{3}{4}\,\frac{g\sqrt{\pi}}{\beta c^2\sqrt{c\beta}} & \mathrm{csgn}(c\beta)=1 \\ \infty & \mathit{otherwise} \end{cases} \tag{4.1.2.2}$$

ここでは，βcが正の場合(csgn(βc)=1)とそれ以外の場合(otherwise)に分けて答えを返している．しかしこのような意図したきれいな結果をいつもMapleが返してくれるとは限らない．これだけしか知らないと，なにかうまくいかないときにお手上げになってしまう．このようなきれいで簡単な結果に行き着く前の，裏でおこなういくつかの予備計算を省略せずに示そう．

先ず鉄則０にしたがってrestartをかけ，関数を定義する．

```
> restart;
  f1:=unapply(x*exp(-beta*c*x^2)*(1+beta*g*x^3),x);
```

$$f1 := x \rightarrow x\,\mathrm{e}^{-\beta c x^2}\left(1+\beta g x^3\right) \tag{4.1.2.3}$$

次には鉄則１にしたがって積分する前にどのような関数かプロットしてみる．そのままplotへ投げると怒られる．

```
> plot(f1(x),x=-10..10);
Warning, unable to evaluate the function to numeric
values in the region; see the plotting command's help
page to ensure the calling sequence is correct
```

これは鉄則２にあるとおり，数値を代入すれば，beta,c,gなどのパラメータの値が入っていないためとわかる．

```
> f1(10);
```

$$10\,\mathrm{e}^{-100\,\beta c}\left(1+1000\,\beta g\right) \tag{4.1.2.4}$$

適当に値を代入する．

```
> c:=1; g:=0.01; beta:=0.1;
```

$$c := 1$$
$$g := 0.01$$
$$\beta := 0.1 \tag{4.1.2.5}$$

再度プロットを試みる．

```
> plot(f1(x),x=-10..10);
```



実際に積分してみる．ここでは，鉄則３にしたがって，式を頭から打ち込むのではなく内側からみていくことが肝要である．これは問題を解いていく時に，思考が必ずたどるであろう順番に相当する．

先ず変数に入れた数値をクリアする．

```
> c:='c': g:='g':beta:='beta':
```

不定積分でこの関数が積分できることを確認する．

```
> int(f1(x),x);
```

(4.1.2.6)

$$-\frac{1}{2\,e^{\beta c x^2}\,\beta c}+\beta g\left(-\frac{1}{2}\,\frac{x^3 e^{-\beta c x^2}}{\beta c}+\frac{3}{2}\,\frac{-\frac{1}{2}\,\frac{x\,e^{-\beta c x^2}}{\beta c}+\frac{1}{4}\,\frac{\sqrt{\pi}\,\mathrm{erf}\left(\sqrt{\beta c}\,x\right)}{\beta c\sqrt{\beta c}}}{\beta c}\right) \tag{4.1.2.6}$$

次にx=-alpha..alphaの定積分を実行する．これは上記のコマンドに付け足すようにしていく．

```
> int(f1(x),x=-alpha..alpha);
```

$$-\frac{1}{4}\,\frac{g\left(4\,\alpha^3\,e^{-\beta c \alpha^2}\,\beta c\sqrt{\beta c}+6\,\alpha\,e^{-\beta c\alpha^2}\sqrt{\beta c}-3\sqrt{\pi}\,\mathrm{erf}\left(\sqrt{\beta c}\,\alpha\right)\right)}{\beta c^2\sqrt{\beta c}} \tag{4.1.2.7}$$

さらにα⇒∞としてみる．

```
> limit(int(f1(x),x=-alpha..alpha),alpha=infinity);
```

$$\lim_{\alpha\to\infty}\left(-\frac{1}{4}\,\frac{1}{\beta c^2\sqrt{\beta c}}\left(g\left(4\,\alpha^3\,e^{-\beta c\alpha^2}\,\beta c\sqrt{\beta c}+6\,\alpha\,e^{-\beta c\alpha^2}\sqrt{\beta c}-3\sqrt{\pi}\,\mathrm{erf}\left(\sqrt{\beta c}\,\alpha\right)\right)\right)\right) \tag{4.1.2.8}$$

ところがこれでは答えを返してくれない．積分した後のそれぞれの項を見るとβ*c>0を仮定すれば簡単になることが分る．assumeを使って，このような変数の仮定おこなう．

```
> assume(beta*c>0);
```

結果として最初に出した解答を得る．

```
> limit(int(f1(x),x=-alpha..alpha),alpha=infinity);
```

$$\frac{3}{4}\,\frac{\sqrt{\pi}\,g}{\beta\sim c\sim^2\sqrt{\beta\sim c\sim}} \tag{4.1.2.9}$$

式のフォローのデフォルト

Mapleで実際に数式をいじる状況というのは，ほとんどの場合が既知の数式変形のフォローだろう．例えば，論文で「(1)式から(2)式への変形は自明である」とかいう文章で済ましている変形が本当にあっているのかを確かめたい時．一番単純なやり方は自明と言われた前後の式が一致していることを確かめるだけで十分である．

最も単純な確認法は以下の通り，変形の前後の式を手入力してその差をexpandした結果が0か否かです．

```
> ex1:=(x-3)^4;
```

$$ex1 := (x-3)^4 \tag{4.1.3.1}$$

```
> ex2:=x^4-12*x^3+54*x^2-108*x+81;
```

$$ex2 := x^4-12\,x^3+54\,x^2-108\,x+81 \tag{4.1.3.2}$$

```
> expand(ex1-ex2);
```

$$0 \tag{4.1.3.3}$$

0ならば式の変形は保証されているので，その導出が間違いでなく誤植なども ないことが確認できる．ただ，これだけでは変形の哲学や技法が身に付くわけではない．あくまでも苦し紛れのデフォルトであることは心に留めておくように．

論理値として確かめたいときには，evalbを使う．

```
> evalb(expand(ex1-ex2)=0);
```

$$\mathit{true} \tag{4.1.3.4}$$

数式処理(演習：1)

いくつかの高校レベルの問題を，Mapleでどのようにして解くかを示す．

例題：式の変形

$x+\dfrac{1}{x}=t$としたとき$x^2+\dfrac{1}{x^2}$をtで表わせ．

解答例

tを代入する．

```
> t:=x+1/x:
```

2乗してみる．

```
> expand(t^2);
```

$$x^2+2+\frac{1}{x^2} \tag{5.1.1.1}$$

結果から明らかに，

```
> expand(t^2)-2;
```

$$x^2+\frac{1}{x^2} \tag{5.1.1.2}$$

したがって，

```
> x^2+1/x^2='t'^2-2;
```

$$x^2+\frac{1}{x^2}=t^2-2 \tag{5.1.1.3}$$

't'はtへの代入を一時的にキャンセルするため．

課題

$x+\dfrac{1}{x}=t$としたとき$x^3+\dfrac{1}{x^3}$をtで表わせ．

課題(旭川大)

$a^2+\dfrac{1}{a^2}=4$のとき，次の値を求めよ．ただし，$1<a<2$とする．

(1) $a^2-\dfrac{1}{a^2}$, (2) $a^3-\dfrac{1}{a^3}$, (3) $a^5-\dfrac{1}{a^5}$

ヒント：(2)の因数分解は以下のようにする．

```
> subs(b=1/a,factor(a^3-b^3));
```

$$\left(a-\frac{1}{a}\right)\left(a^2+1+\frac{1}{a^2}\right) \tag{5.1.3.1}$$

解答例

```
> restart;
> t:=a+1/a;
```

$$t:=a+\frac{1}{a} \tag{5.1.3.1.1}$$

```
> solve(expand(z^2-2)=4,z);
```

$$\sqrt{6},\ -\sqrt{6} \tag{5.1.3.1.2}$$

```
> ans1:=solve(expand((a+1/a)^2-2=4),a);
```

$$ans1:=-\frac{1}{2}\sqrt{6}-\frac{1}{2}\sqrt{2},\ \frac{1}{2}\sqrt{6}+\frac{1}{2}\sqrt{2},\ -\frac{1}{2}\sqrt{6}+\frac{1}{2}\sqrt{2},\ \frac{1}{2}\sqrt{6}-\frac{1}{2}\sqrt{2} \tag{5.1.3.1.3}$$

```
> evalf(ans1);
```

$$-1.931851653,\ 1.931851653,\ -0.5176380910,\ 0.5176380910 \tag{5.1.3.1.4}$$

```
> a1:=ans1[2];
```

$$a1:=\frac{1}{2}\sqrt{6}+\frac{1}{2}\sqrt{2} \tag{5.1.3.1.5}$$

```
> eq1:=normal(subs(a=a1,a^2-1/a^2));
```

$$eq1:=\frac{8\left(3+\sqrt{6}\sqrt{2}\right)}{\left(\sqrt{6}+\sqrt{2}\right)^2} \tag{5.1.3.1.6}$$

```
> simplify(expand(rationalize(eq1)));
```

$$2\sqrt{3} \tag{5.1.3.1.7}$$

例題：

2つの2次方程式$x^2-(k+1)x-k^2=0,\ x^2-\dfrac{1}{2}kx-k=0$ がただひとつの共通な実数解をもつとき，定数kの値とその時の共通解を求めよ．

解答例

2つの方程式を代入する．

```
> eq1:=x^2-(k+1)*x-k^2=0:
  eq2:=x^2-1/2*k*x-k=0:
```

{x,k}を変数とする2つの方程式とみなして連立方程式を解く．

```
> solve({eq1,eq2},{x,k});
```

$$\{x=0,k=0\},\ \{x=-1,k=2\},\ \left\{k=-\frac{1}{2}+\frac{3}{2}\,RootOf\left(5\_Z^2-2\_Z+1,\ label=\_L8\right),\ x=2\,RootOf\left(5\_Z^2-2\_Z+1,\ label=\_L8\right)\right\} \tag{5.2.1.1}$$

実数解はk=0, k=2のときにx=0, x=-1である．k=2の時の2つの方程式の左辺を2次関数とみなしてプロットしてみる．k=0も同様．

```
> plot(subs(k=2,{lhs(eq1),lhs(eq2)}),x=-2..0);
```



課題

2つの2次方程式 $-x^2+(2+k)x-2k=0,\ x^2+(-1-k)x+k=0$ がただひとつの共通な実数解をもつとき，定数kの値とその時の共通解を求めよ．

課題

$\frac{1}{3-\sqrt{7}}$の整数部分をp, 小数部分をqとしたときの，$p^2+2pq+4q^2$を求めよ．

ヒント：trunc, frac等を調べよ．

数式処理(演習：2)

例題：1次変換と直線

行列A= $\begin{bmatrix} a & 3 \\ 4 & b \end{bmatrix}$ *の表わす1次変換fによって，直線 2x-y-2=0が直線 3x-4y+10=0 に移されるとき，a, b の値を求めよ．*

解答例

```
> restart;with(LinearAlgebra):
A:=Matrix(2,2,[[a,3],[4,b]]);
```

$$A := \begin{bmatrix} a & 3 \\ 4 & b \end{bmatrix} \quad (6.1.1.1)$$

直線 2x-y-2 = 0 上の点を媒介変数tを用いて表わす．

```
> X:=Vector([t,2*t-2]);
```

$$X := \begin{bmatrix} t \\ 2t-2 \end{bmatrix} \quad (6.1.1.2)$$

行列によって変換された後の点(x',y')は,

```
> (A.X);
```

$$\begin{bmatrix} at+6t-6 \\ 4t+b(2t-2) \end{bmatrix} \quad (6.1.1.3)$$

この点が直線 3x-4y+10=0 上にあるから，先程求めた，(x',y')を代入する．

```
> Eq1:=3*(A.X)[1]-4*(A.X)[2]+10=0;
```

$$Eq1 := 3at+2t-8-4b(2t-2)=0 \quad (6.1.1.4)$$

tについて整理すると,

```
> collect(Eq1,t);
```

$$(3a+2-8b)t-8+8b=0 \quad (6.1.1.5)$$

これがtによらずに成立するためには，恒等式でなければならない．tの0，1次の係数を取り出す．

```
> Eq2:={coeff(lhs(Eq1),t,1)=0,
coeff(lhs(Eq1),t,0)=0};
```

$$Eq2 := \{3a+2-8b=0, -8+8b=0\} \quad (6.1.1.6)$$

a,bについて解く．

```
> solve(Eq2,{a,b});
```

$$\{b=1, a=2\} \quad (6.1.1.7)$$

課題

行列A= $\begin{bmatrix} 2 & 3 \\ 4 & 1 \end{bmatrix}$ の表わす1次変換fによって，自分自身に移る直線の方程式を求めよ．

課題：行列の対角化

1 $A=\begin{bmatrix} 1 & 2 \\ -1 & 4 \end{bmatrix}$ *の固有値を求めよ．*

2 固有ベクトルで作られる行列をPとおくとき，*P^(-1).A.P*を求めよ．

3 *A^n* (*n*は自然数)を求めよ．

例題：行列のスペクトル分解

一般に，2次の正方行列Aが異なる2つの固有値a,bを持つとき,

aP + bQ=A, P+Q=E

を満たす行列P,Qに対して，次の事が成り立つ．

（1）　PQ=QP=0,

（2）　P^2=P,

（3）　Q^2=Q

上のように行列を分解することをスペクトル分解という．

また，A^n=(aP+bQ)^n = a^nP+ b^nQ

となる．

$A=\begin{bmatrix} 4 & 2 \\ 1 & 3 \end{bmatrix}$ *に対して，行列P, Qを求めよ．*

解答例

行列のまま変形するのが難しそうなので解答例を示す．みそは，「初めは，未知変数とみなして連立方程式を解いておいて，後から，行列要素を入れる(subs)．」

```
> restart;
with(LinearAlgebra):
A:=Matrix(2,2,[[4,2],[1,3]]);
```

$$A := \begin{bmatrix} 4 & 2 \\ 1 & 3 \end{bmatrix} \quad (6.3.1.1)$$

```
> l,V:=Eigenvectors(A);
```

$$l, V := \begin{bmatrix} 2 \\ 5 \end{bmatrix}, \begin{bmatrix} -1 & 2 \\ 1 & 1 \end{bmatrix} \quad (6.3.1.2)$$

```
> s1:=solve({l[1]*P+l[2]*Q=AA,P+Q=EE},{P,Q});
```

$$s1 := \left\{P=\frac{5}{3}EE-\frac{1}{3}AA, Q=-\frac{2}{3}EE+\frac{1}{3}AA\right\} \quad (6.3.1.3)$$

```
> E:=Matrix(2,2,shape=identity);
P:=simplify(subs({EE=E,AA=A},rhs(s1[1])));
```

$$E := \begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}$$

$$P := \begin{bmatrix} \frac{1}{3} & -\frac{2}{3} \\ -\frac{1}{3} & \frac{2}{3} \end{bmatrix} \quad (6.3.1.4)$$

以下略

課題

$A=\begin{bmatrix} 4 & 2 \\ 1 & 3 \end{bmatrix}$ *に対して，スペクトル分解した行列P, Q を求めよ．*

また，$PQ = QP = 0,\ P^n = P,\ Q^n = Q$が成り立つことを確かめよ．
さらにAの逆行列を$cP + dQ$と表わしたとき，実数c, dを求めよ．

課題：微分

2曲線 $y = x^3 - 2x + 1,\ y = x^2 + 2ax + 1$ が接するとき，aの値を求めよ．また，その接点における共通の接線の方程式を求めよ．

課題：積分

関数$f(x)$が$f(x) = 2x + \int_0^1 (x+t)f(t)\,dt$を満たすとき，$f(x)$を求めよ．